
Introduction

The computer package Itovsn3 was initially developed for the computer algebra
package REDUCE (Kendall 1987, 1988, 1991c), and then re-implemented in
Mathematica Kendall (1993b) and then in AXIOM Kendall (2001) using the
information-hiding features of these packages. In this research report we present
seven Mathematica notebooks which use Itovsn3. The intention is to provide
material for people who want to understand Itovsn3 well enough to use it for
their own purposes.

The notebooks in question are as follows.

• Intro.nb: this introduces the Itovsn3 constructs and gives a simple ex-
ample of programming to compute Itô integrals;

• Reference.nb: a non-executable notebook displaying the main contents
of the Itovsn3 package in a convenient reference form;

• Bessel.nb: calculations with the three-dimensional Bessel process, up to
computation of hitting probabilities and heat kernel;

• Itoarea.nb: calculation of the moment generating function of Lévy stochas-
tic area following an idea due to Svante Jansson;

• BlackScholes.nb: using Itovsn3 to derive the celebrated Black-Scholes
formula from mathematical finance;

• MardiaDryden.nb: derivation of the Mardia-Dryden distribution from sta-
tistical shape theory;

• Reflect.nb: a notebook showing the original computer algebra argument
for an elegant and helpful representation for reflection-coupled Brownian
motions in a half-plane.

As an appendix there is a listing of the current Itovsn3 package.
The format of the report is as follows: separate sections contain hyperlinks

to each Mathematica notebooks, each with a preliminary paragraph describing
the basic features of the notebook and supplying extra references. (If you have
access to Mathematica then you may choose to configure your browser to launch
the notebooks on demand; but note you will need also to download Itovsn3.m
and place it on your path!) The report concludes with a brief summary.

The terminology “Itovsn3” arises by combining “Itô” (in honour of Kiyosi
Itô) and a contraction of “version 3”, at which version level package development
stabilized . . . .

1 Intro.nb: an introductory notebook

The first part of this notebook is concerned with various ways of introducing
semimartingales, ranging from the most general through to specific procedures
for introduction of standard Brownian motions. Notice in particular the versa-
tile and useful Itosde procedure, which corresponds exactly to definition of a
semimartingale using a stochastic differential equation. In the example Itosde
is used to define Z as the solution to

dIZ = Y dIX −X dIY (1.1)

so that Z is (proportional to) the Itô stochastic area corresponding to the two-
dimensional Brownian motion (X, Y ).

In the second part of the notebook there is a demonstration of simple pro-
gramming using Itovsn3. An algorithm is implemented which rewrites Itô inte-
grals involving polynomials of time t and a simple Brownian motion B, rewriting
them in terms of classical integrals using only dt and not dIB (though of course
the integrands still may contain instances of B!). A partial check of the algo-
rithm is conducted using the famous relationship of Itô integrals to Hermite
polynomials.

Link to Intro.nb
Note that Itovsn3.m is needed for this and the other notebooks.

Link to Itovsn3.m

Intro.nb
Itovsn3.m


2 Reference.nb: reference to Itovsn3 features

This notebook is completely inactive, but uses the “folding” feature of Math-
ematica notebooks (the hidden cell feature) to produce a convenient on-line
reference to the construction of Itovsn3. A list is given of all procedures of
Itovsn3, grouped thematically in sections. Associated hidden calls contain indi-
cations of useage, brief textual notes, and listings of some of the relevant source
code.

Link to Reference.nb

3 Bessel.nb

The Bessel process is the simplest non-trivial variation of the theme of stan-
dard Brownian motion. The Bessel process R of k degrees of freedom, often
written BESS(k), is most readily defined for positive integers k by a distance
representation: R is distributed as the Euclidean distance from the origin of
a k-dimensional Brownian motion (each of the k coordinates is an indepen-
dent standard Brownian motion). A special case (k = 3) is considered in this
notebook.

From this representation one can use stochastic calculus to deduce the gen-
eral stochastic differential equation

dIR = dIW +
k − 1
2R

dt (3.1)

which can be used to define the Bessel process for positive real values of the
“dimension” k.

In the notebook we compute hitting probabilities and transition kernel for a
BESS(3) process. Much of this generalizes to other dimensions (though closed
forms for the transition kernel can be expected only for odd integral k). How-
ever automatic derivation of Equation (3.1) for general integer k is in fact a
complicated matter for computer algebra packages; it requires computation us-
ing “symbolic dimension” k. Eventually this leads to interesting, complicated,
and not completely resolved issues concerning automatic simplification. See
Kendall (1990) for a way of carrying out this sort of computation in Itovsn3,
and Kendall (1992) for similar but more involved computations in statistical
asymptotics, and a discussion of the simplification issues involved.

More advanced investigation of Bessel processes using Itovsn3 may be found
in Kendall (1991b).

Link to Bessel.nb

Reference.nb
Bessel.nb


4 Itoarea.nb: calculation of the moment gener-
ating function of the Itô stochastic area

Paul Lévy derived a beautiful formula for the moment generating function of
the Itô stochastic area defined in Equation (1.1). Svante Jansson developed a
method of proving this using stochastic calculus, as described in Protter (1990,
Theorem 42), and a treatment using REDUCE Itovsn3 is to be found in Kendall
(1993a). This notebook translates the REDUCE version to Mathematica and
exploits Mathematica’s DSolve facility for solving the relevant differential equa-
tions.

Link to Itoarea.nb

5 BlackScholes.nb: pricing options

Financial mathematics is a natural application area for Itovsn3, not as yet
exploited. The original Mathematica implementation of Itovsn3 was initially
applied in Kendall (1993b) to provide a computer algebra treatment of a hedging
problem originally solved by Duffie and Richardson, and in the same volume a
diffusion-based implementation of stochastic calculus, Steele and Stine (1993),
was used to derived the celebrated Black-Scholes option pricing formula. Here
we give an Itovsn3 treatment of Black-Scholes. A derivation using the AXIOM
implementation is to be found in Kendall (1998a).

Link to BlackScholes.nb

6 MardiaDryden.nb: statistics of shape

The initial application area of the original REDUCE-based implementation of
Itovsn3 was to the statistical theory of shape as expounded in Stoyan, Kendall,
and Mecke (1995), Small (1996), Dryden and Mardia (1998), Barden, Carne,
Kendall, and Le (1999). The original work (Kendall 1987, 1988) is still being
developed, for example in Kendall (1998b). Here we use stochastic differential
equation theory to derive the Mardia-Dryden distribution from shape theory
(Mardia and Dryden (1989)). Note that there are other stochastic calculus
approaches to this, for example Kendall (1991a), Le (1991). The attraction of
the approach described here is that it adopts a rather empirical search method
based loosely on heat-kernel asymptotics. A similar method is used to derive
approximations to a shape distribution in Kendall (1998b).

An implementation of automatic diffusion geometry within the REDUCE
version of Itovsn3 has been used to give very rapid derivations of some shape
geometries: see Kendall (1991c).

Link to MardiaDryden.nb

Itoarea.nb
BlackScholes.nb
MardiaDryden.nb


7 Reflect.nb: a coupling example

The final example is a genuine research contribution to the theory of coupling
of random processes. In the course of the investigation reported in (Burdzy and
Kendall 2000), motivated by problems in perfect simulation (Propp and Wilson
(1996), Kendall (1998c)), it was required to gain insight into the behaviour of
two reflection-coupled Brownian motions evolving within a half-plane with re-
flecting boundary conditions. An elegant and simple representation was derived
using the REDUCE implementation of Itovsn3 as an exploratory tool. In fact
the representation is obvious when considered with the advantage of hindsight;
nevertheless its discovery is directly attributable to Itovsn3.

A more extensive and substantial approach to a coupling problem, using a
combination of algebraic, graphical, and numerical techniques all controlled by
REDUCE Itovsn3, is reported in Arous, Cranston, and Kendall (1995).

Link to Reflect.nb

8 Conclusion

It is interesting to note the common features of these notebooks. Many of
them (Bessel.nb, BlackScholes.nb, Itoarea.nb, MardiaDryden.nb, ) can be
viewed as using semimartingale stochastic calculus, particularly local martingale
theory, to find solutions to stochastic differential equations. However this does
not exhaust the possibilities of Itovsn3; consider that in other cases (Intro.nb,
Reflect.nb) it is used to construct algorithms for solving stochastic calculus
problems, or to find special and informative representations of geometrically-
defined semimartingales. The power and attraction of Itovsn3 lies in the way in
which it relates the computer algebra “front-end” (the way in which expressions
are input into the package) very closely to a familiar, expressive, and powerful
stochastic calculus formalism. The AXIOM implementation of (Kendall 1998a,
2001) carries this further still, by using AXIOM ’s facilities for programming
mathematical structure so as to implement the actual algebra of stochastic dif-
ferentials: it would be a very useful programming exercise now to introduce
these ideas into Itovsn3 itself — though after such an upgrade it had better be
re-named Itovsn4!

Reflect.nb

A The package Itovsn3.m
(*

Itovsn3.m: a Mathematica package for
Symbolic Ito calculus Itovsn3. Version 3.70,

Copyright March, June 1992, April 1998, November 2002, January 2003
Author: Wilfrid S.Kendall.

Version 3.68 was tested on
Mathematica version 1, Macintosh, Windows 3.1;
Mathematica version 2, Windows 3.1;
Mathematica version 2.2.3, Windows 97;
Mathematica version 3.0, Linux.

Version 3.39 has been tested on
Mathematica version 4.0.0.0, Windows 2000

The author disclaims any responsibility for problems experienced by
users as a result of using this software, but welcomes your feedback!
The software may not be copied or distributed for commercial gain
without written permission from the author, but may be passed to
others if left intact in its present form.

Comments please to w.s.kendall@warwick.ac.uk or to

Prof W S Kendall,
Department of Statistics,
University of Warwick,
Coventry CV4 7AL, UK.
*)

BeginPackage["Itovsn3‘"]

AddDrift::usage =
"AddDrift[dX,DriftdX] sets Drftbydt[dX] = DriftdX/dt, so that\n
(for example) Drift[dX] yields DriftdX."

AddFixed::usage =
"AddFixed[t0,X,X0] sets Fixed[t0,X] = X0, so that\n
(for example) InitialValue[t0,X] yields X0."

AddQuadVar::usage =
"AddQuadVar[dX,dY,xdt] sets Brktbydt[dX,dY] = xdt/dt, altering \n
the effect of substitution using ItoMultiplications, so that\n
(for example) ItoExpand[dX dY] yields xdt. Similarly for\n
AddQuadVar[dX dY,xdt] and AddQuadVar[dX^2,xdt]."

Brktbydt::usage =
"Brktbydt[dX,dY] is a placeholder for the FORMAL quotient by dt of\n
the bracket differential dX dY."

BrownBasis::usage =
"BrownBasis[SemimartingaleList,InitialValueList] introduces and sets\n
up second- and first-order structure for an independent set of Brownian\n
basic semimartingales, identifiers in SemimartingaleList, with initial \n
value expressions given by InitialValueList. It uses BrownSingle[X,X0] \n
as a supplementary procedure. Corresponding basic stochastic differential\n
identifiers are created by prepending ’d’ to the SemimartingaleList \n
identifiers."

BrownSingle::usage =
"BrownSingle[X,X0] introduces and sets up a single Brownian basic\n
semimartingale identifier X (creating the semimartingale differential\n
identifier dX) with initial value expression X0."

Drftbydt::usage =
"Drftbydt[dX] is a placeholder for the FORMAL quotient by dt
of Drift of dX.\n"



Drift::usage =
"Drift[sd] computes the drift differential of the stochastic\n
differential expression sd. NOTE that Drift assumes that sd is\n
genuinely a stochastic differential expression."

Fixed::usage =
"Fixed[t0,X] is a placeholder for the fixed value of the basic \n
semimartingale X at time t0 (usually t0=0)."

InitialValue::usage =
"InitialValue[t0,f] computes the value of the expression f \n
at time t0."

Introduce::usage =
"Introduce[Smgl,dSmgl] introduces the basic semimartingale identifier \n
Smgl with associated basic stochastic differential identifer dSmgl. \n
Attempts to reintroduce semimartingale or stochastic differential\n
identifiers are reported as errors."

ItoD::usage =
"ItoD[f] computes the stochastic differential of the semimartingale\n
expression f. NOTE that ItoD assumes that f is a genuine\n
semimartingale expression!"

ItoExpand::usage =
"ItoExpand[sd] computes simplication of the stochastic differential\n
products in the expression sd which is its argument."

ItoInit::usage =
"ItoInit[t,dt] starts things off with basic structures, using the \n
identifier t for time variable and the identifier dt for its \n
differential."

ItoIntegral::usage =
"ItoIntegral[sd] represents the Ito integral of the stochastic\n
differential expression sd."

ItoReset::usage =
"ItoReset[t,dt] resets all structures, using ItoInit[t,dt]."

Itosde::usage =
"Itosde[X,dX==sd,X0] introduces and sets up a basic semimartingale \n
identifier X with basic stochastic differential identifier dX and \n
initial value expression X0, and satisfying the second- and first- \n
order structure implied by the stochastic differential equation dX==sd."

ItoStatus::usage =
"ItoStatus[] reports current structures."

RandomQ::usage =
"RandomQ[x,sdl]==True if x is an expression in semimartingales or\n
stochastic differentials excluding those given in sdl."

BSDQ::usage =
"BSDQ[x,sdl]==True if x is a basic stochastic differential\n
excluding those given in sdl."

(* ============================= *)

Begin["‘private‘"]

(* =>Implementing the Ito formula: *)
ItoD[f_] := Block[
{ ff =(f/.ItoIntegral[sdx_]->ItoIntegral[sdx,t]) },
ItoExpand[

(Dt[ff,t] dt + (1/2) Dt[ff,{t,2}] dt^2)/.Freeze[sd_]->sd ]
];



ItoExpand[sd_] := ((Expand[sd]
/. ItoIntegral[sdx_,t]->ItoIntegral[sdx])
/. ItoMultiplications);

(* =>Implementing Drift: *)
Drift[sd_] := Apply[Plus,Map[Coefficient[Expand[sd],#] Drftbydt[#] dt &,CSD]];

(* =>Updating first- and second-order structure: *)
AddQuadVar[dX_ dY_,xdt_] := AddQuadVar[dX,dY,xdt];
AddQuadVar[dX_^2,xdt_] := AddQuadVar[dX,dX,xdt];
AddQuadVar[dX_,dY_,xdt_] := (Brktbydt[dX,dY] = xdt/dt);
AddDrift[dX_,DriftdX_] := (Drftbydt[dX] = DriftdX/dt);

(* =>Finding initial value: *)
AddFixed[t0_,y_,y0_] := (Fixed[t0,y]=y0);
Derivative[1,0][Fixed][t0_,x_] := 0;
Derivative[0,1][Fixed][t0_,x_] := 0;
InitialValue[t0_,x_] :=
((((x /. Map[#->MayFix[t0,#]&,CS])
/. ItoIntegral[y_] ->MayFix[t0,ItoIntegral[y]])
/. MayFix[a_,y_] -> MayFix[a,(y/.MayFix[t0,u_]->u)])
/. MayFix[t0,u_] -> Fixed[t0,u]);

(* =>Simplest properties of ItoIntegral: *)
Derivative[0,2][ItoIntegral][sd_,t_] := 0;
Derivative[0,1][ItoIntegral][sd_,t_] := Freeze[sd]/ItoD[t];
Derivative[1][Freeze][sd_] := 0;
Derivative[1,0][ItoIntegral][sd_,t_] := 0;

(* =>Introducing basic semimartingale
and associated stochastic differential: *)
Introduce[X_, dX_] :=
If[Not[MemberQ[CS,X]] && Not[MemberQ[CSD,dX]],
(
X/: Dt[X,t] = dX/dt;
dX/: Dt[dX,t] = 0 ;

ItoIntegral[dX] = X - Fixed[0,X];
CSD = Prepend[CSD,dX];
CS = Prepend[CS,X];
ItoMultiplications = Join[
Map[(dX # -> Brktbydt[dX,#] dt)&,CSD],
ItoMultiplications ];
AddQuadVar[dX dt,0];
CSD
), "Attempt to re-introduce semimartingale or stochastic differential!"
];

(* =>Initialization: *)
ItoInit[time_,dtime_] :=
(
t = time;
dt = dtime;
SetAttributes[Brktbydt,Orderless];
CS = {};
CSD = {};
ItoMultiplications = {};
Introduce[t,dt];
AddDrift[dt,dt];
ItoIntegral[0] = 0;
Fixed[t0_,t] = t0;
Fixed[0,ItoIntegral[y_]] = 0;
TableForm[{
"Itovsn3 initialized",
SequenceForm["with time semimartingale ",t],
SequenceForm["and time differential ",dt]
}]
);



(* =>Resetting stochastic calculus structures: *)
ItoReset[time_,dtime_] :=
(
Map[ItoClear[#,ItoD[#]]&,CS];
Clear[Fixed];
Clear[Brktbydt];
Clear[Drftbydt];

Clear[ItoIntegral];
TableForm[Prepend[First[ItoInit[time,dtime]],"Itovsn3 resetting ..."]]
);

(* =>Test if x involves semimartingales or differentials except
those given in the second list (presumed deterministic!)*)

RandomQ[x_,sdl_] :=
Not[FreeQ[x,Apply[Alternatives,Complement[Union[CS,CSD],sdl]]]]

(* =>Test if x is a basic stochastic differential excepting
those given in the second list*)

BSDQ[x_,sdl_] :=
MemberQ[Complement[CSD,sdl],x]

(* =>Clear aspects of semimartingale X and differential dX: *)
ItoClear[X_,dX_] :=
( dX/: Dt[dX,t] = .; X/: Dt[X,t] = .; );

(* =>Reporting status: *)
ItoStatus[] :=
Print[ColumnForm[{

"---------------",
"Summary of current structure of stochastic differentials",
"- - - - - - - -",
"Current second-order structure of semimartingale differentials:",
TableForm[Outer[ItoExpand[#1 #2]&,CSD,CSD],TableHeadings->{CSD,CSD}],
"- - - - - - - -",
"Current first-order structure of semimartingale differentials:",
TableForm[Map[Drift[#]&,CSD],

TableDirections->Row,TableHeadings->{CSD,{"Drifts:"}}],
"- - - - - - - -",
"Current initial values:",
TableForm[Map[Fixed[0,#]&,CS],
TableDirections->Row,TableHeadings->{CS,{"Initially:"}}],
"---------------"
}]];

(* =>Brownian Basis: *)
BrownBasis[SemimartingaleList_,InitialValueList_] :=
(
Map[Apply[BrownSingle,#]&,
Transpose[{SemimartingaleList,InitialValueList}]];
BrownPairs[Map[ItoD,SemimartingaleList]]
);
BrownSingle[X_,X0_] :=
Block[
{dX=ToExpression[StringJoin["d",ToString[X]]]},
Introduce[X,dX];
AddQuadVar[dX^2,dt];
AddDrift[dX,0];
Fixed[0,X]=X0
];
BrownPairs[SL_] :=
If[Length[SL]>1,
Map[AddQuadVar[First[SL] #,0]&,Rest[SL]];BrownPairs[Rest[SL]]];

(* => Definition using Ito stochastic differential equations: *)
Itosde[X_,dX_==sd_,X0_] :=
(
Introduce[X,dX];
AddDrift[dX,Drift[sd]];



Fixed[0,X]=X0;
AddQuadVar[dX^2,ItoExpand[sd^2]];
Map[AddQuadVar[dX #,ItoExpand[sd #]]&,CSD];
);

End[]
(* ============================= *)

EndPackage[]
(* ============================= *)

